

	******************************************************************
	**
	**
	**		NAME:	ELITE PERSISTENCE IN THE ERA OF EXPANDING OVERSEAS TRADE
	**
	**		MAKES:  Figures 2 (a) and (b)		  		       
	**		
	**
	******************************************************************
	
	
	
	
	

			

		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* global for input/output
*-------------------------------------------------------------------------------


* input
*------

global input = "" 	
	
	
* output
*-------

global output = ""	
		
	
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* preliminaries
*-------------------------------------------------------------------------------


* preliminaries
*--------------

clear
set more off






			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* data
*-------------------------------------------------------------------------------


* use
*----

cd ${input}
use analysis_vars_parl.dta, clear 




			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* assert
*-------------------------------------------------------------------------------


* assert
*-------

assert 	parl_year >= 1536 & parl_year <= 1760


duplicates tag parl_group, gen(tt)
assert tt == 0 if parl_group != .
drop tt






			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* globals
*-------------------------------------------------------------------------------


* globals
*--------

global yr1 		= "parl_year > 1553 & parl_year < 1639"
global yr2 		= "parl_year >= 1660 & parl_year <= 1756"

global yr3 		= "parl_year >= 1639 & parl_year <= 1649"
global yr4 		= "parl_year >= 1628 & parl_year <= 1660"

global yr5 		= "parl_year < 1639 & parl_year >= 1583"
global yr6 		= "parl_year > 1550 & parl_year <= 1750"

global yr7 		= "parl_year > 1553 & parl_year <= 1640"


global xline1	= "xline(1660, lcolor(gs12))"
global xline2	= "xline(1688, lcolor(gs12) lpattern(line) )"
global xline3	= "xline(1640, lcolor(gs12))"
global xline4	= "xline(1562, lcolor(gs12)  lpattern(line) )"


	
	

		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**


*-------------------------------------------------------------------------------
* PLOT:  trend in total commercial   (FIGURE 2 (a))
*-------------------------------------------------------------------------------


* hi lo
*------

capture drop hi lo
gen hi = 31
gen lo = 0
sort parl_group



* this plot
*----------

global outcome1 	= "int_commtot_1_pperc"
global outcome2 	= "int_trader_pperc"
global label = "all constituencies"
global scale = "0(10)30"

gen ${outcome1}_pct = ${outcome1} * 100
gen ${outcome2}_pct = ${outcome2} * 100

* plot
*-----

# delimit ;

twoway 
	
	
		// shaded area
	(rarea hi lo parl_year	
		if   ${yr3} 
		,  color(gs12) )
		
		

	(lpoly ${outcome1}_pct parl_year
		if   ${yr1}		
		,  bwidth(7) lcolor(gs1)  lwidth(medthick) lpattern(line)  ${xline2} ${xline4} )
	(lpoly ${outcome1}_pct parl_year
		if   ${yr2}		
		,  bwidth(7) lcolor(gs1)  lwidth(medthick) lpattern(line)  )
		
	(lpoly ${outcome2}_pct parl_year
		if   ${yr1}		
		,  bwidth(7) lcolor(gs5)  lwidth(medthick) lpattern(shortdash)  ${xline2} ${xline4} )
	(lpoly ${outcome2}_pct parl_year
		if   ${yr2}		
		,  bwidth(7) lcolor(gs5)  lwidth(medthick) lpattern(shortdash)  )
		
		
	(scatter ${outcome1}_pct parl_year
		if    ${yr1}		
		,  mcolor(gs1) msize(medlarge) msymbol(circle))
	(scatter ${outcome1}_pct parl_year
		if    ${yr2}			
		,  mcolor(gs1) msize(medlarge) msymbol(circle))		
		
	(scatter ${outcome2}_pct parl_year
		if    ${yr1}		
		,  mcolor(gs5) msize(medlarge) msymbol(circle_hollow))
	(scatter ${outcome2}_pct parl_year
		if    ${yr2}			
		,  mcolor(gs5) msize(medlarge) msymbol(circle_hollow))		
		
		,
		yscale(noline)
		xscale(noline)
		
		
		xsize(8.5)
		ysize(3)

		title("", color(black) size(medsmall) pos(11) ) 
		xlab( , 
			labsize(medsmall) nogrid) 
		ylab( ${scale}, 
			angle(hori) labsize(medsmall) nogrid  )
		xtitle("Year" , 
			color(black) size(large) )
		ytitle("Percentage of MPs", 
			color(black) size(large)  )
		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(none) lcolor(white) ilstyle(none)) 
		legend(order(
				2 "% MPs w/ any Commercial Occupation"
				4 "% MPs w/ Overseas Atlantic Trade Occupation"
				)
			cols(2)
			pos(6)
			
			region( color(none) )
			size(large)
			)
		
		;

# delimit cr


	* export graph
	*-------------

	cd "${output}"
	graph export "trend_commercial.pdf", replace
	
	
	
	


		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**


*-------------------------------------------------------------------------------
* PLOT:  trend in total passive commercial (FIGURE 2 (b))
*-------------------------------------------------------------------------------
	
	

* this plot
*----------

capture drop hi lo
gen hi = 31
gen lo = 0
sort parl_group

global outcome1 	= "pass_anytrade_i_pperc"
global label = "all constituencies"
global scale = "0(10)30"

gen ${outcome1}_pct = ${outcome1} * 100


* plot
*-----
set more off
# delimit ;

twoway 
	
	
		// shaded area
	(rarea hi lo parl_year	
		if   ${yr3} 
		,  color(gs12) )		

	(lpoly ${outcome1}_pct parl_year
		if   ${yr1}		
		,  bwidth(7) lcolor(gs1)  lwidth(medthick) lpattern(line)  ${xline2} ${xline4} )
	(lpoly ${outcome1}_pct parl_year
		if   ${yr2}		
		,  bwidth(7) lcolor(gs1)  lwidth(medthick) lpattern(line)  )
	
	(scatter ${outcome1}_pct parl_year
		if    ${yr1}		
		,  mcolor(gs1) msize(medlarge) msymbol(circle))
	(scatter ${outcome1}_pct parl_year
		if    ${yr2}			
		,  mcolor(gs1) msize(medlarge) msymbol(circle))		
		,
		yscale(noline)
		xscale(noline)
	
		xsize(8.5)
		ysize(3)
		
		title("", color(black) size(medsmall) pos(11) ) 
		xlab( , 
			labsize(medsmall) nogrid) 
		ylab( ${scale}, 
			angle(hori) labsize(medsmall) nogrid  )
		xtitle("Year" , 
			color(black) size(large) )
		ytitle("Percentage of MPs", 
			color(black) size(large)  )
		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(none) lcolor(white) ilstyle(none)) 
		legend(order(
				2 "% MPs w/ a Financial Investment in Overseas Trade"
				)
			cols(2)
			pos(6)			
			region( color(none) )
			size(large)
			)
		;

# delimit cr


	* export graph
	*-------------

	cd "${output}"
	graph export "trend_passive.pdf", replace	
	
	
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	

						** end of do file ** 	
	
	
	